# $Id$
# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=perl-netaddr-ip
pkgver=4.079
pkgrel=6
pkgdesc="Perl module to manage IP addresses and subnets"
arch=('x86_64')
url="http://search.cpan.org/dist/NetAddr-IP/"
license=('PerlArtistic' 'GPL')
depends=('perl')
options=('!emptydirs')
source=(http://search.cpan.org/CPAN/authors/id/M/MI/MIKER/NetAddr-IP-${pkgver}.tar.gz)
md5sums=('990dfcbffae356835c536e8ab56a2880')

build() {
  cd NetAddr-IP-${pkgver}
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd NetAddr-IP-${pkgver}
  make test
}

package() {
  cd NetAddr-IP-${pkgver}
  make DESTDIR="${pkgdir}" install
}
